

## 2.2 Machine-Dependent Assembler Features (SIC/XE Assembler)

Instruction Formats, Addressing Modes, and  
Program Relocation

# SIC/XE Assembly Program(fig.2.5)

5	COPY	<u>START</u>	<u>0</u>	COPY FILE FROM INPUT TO OUTPUT
10	FIRST	STL	RETADR	SAVE RETURN ADDRESS
12		<u>LDB</u>	<u>#LENGTH</u>	ESTABLISH BASE REGISTER
13		<u>BASE</u>	<u>LENGTH</u>	
15	CLOOP	+JSUB	RDREC	READ INPUT RECORD
20		LDA	LENGTH	TEST FOR EOF (LENGTH = 0)
25		COMP	#0	extended format
30		JEQ	ENDFIL	EXIT IF EOF FOUND
35		+JSUB	WRREC	WRITE OUTPUT RECORD
40		J	CLOOP	LOOP
45	ENDFIL	LDA	EOF	INSERT END OF FILE MARKER
50		STA	BUFFER	
55		LDA	#3	SET LENGTH = 3
60		STA	LENGTH	immediate addressing
65		+JSUB	WRREC	WRITE EOF
70		J	@RETADR	RETURN TO CALLER
80	EOF	BYTE	C'EOF'	indirect addressing
95	RETADR	RESW	1	
100	LENGTH	RESW	1	LENGTH OF RECORD
105	BUFFER	RESB	4096	4096-BYTE BUFFER AREA
110	.			

# SIC/XE Assembly Program

```
115      .          SUBROUTINE TO READ RECORD INTO BUFFER
120      .
125      RDREC      CLEAR      X          CLEAR LOOP COUNTER
130                  CLEAR      A          CLEAR A TO ZERO
132                  CLEAR      S          CLEAR S TO ZERO
133                  +LDT        #4096
135      RLOOP      TD          INPUT      TEST INPUT DEVICE
140                  JEQ         RLOOP     LOOP UNTIL READY
145                  RD          INPUT     READ CHARACTER INTO REGISTER A
150                  COMPR       A,S       TEST FOR END OF RECORD (X'00')
155                  JEQ         EXIT      EXIT LOOP IF EOR
160                  STCH        BUFFER,X  STORE CHARACTER IN BUFFER
165                  TIXR        T         LOOP UNLESS MAX LENGTH
170                  JLT         RLOOP     HAS BEEN REACHED
175      EXIT       STX         LENGTH     SAVE RECORD LENGTH
180                  RSUB
185      INPUT      BYTE       X'F1'      CODE FOR INPUT DEVICE
195      .
```

# SIC/XE Assembly Program

```
200      .      SUBROUTINE TO WRITE RECORD FROM BUFFER
205      .
210      WRREC      CLEAR      X      CLEAR LOOP COUNTER
212      .      LDT      LENGTH
215      WLOOP      TD      OUTPUT      TEST OUTPUT DEVICE
220      .      JEQ      WLOOP      LOOP UNTIL READY
225      .      LDCH      BUFFER,X      GET CHARACTER FROM BUFFER
230      .      WD      OUTPUT      WRITE CHARACTER
235      .      TIXR      T      LOOP UNTIL ALL CHARACTERS
240      .      JLT      WLOOP      HAVE BEEN WRITTEN
245      .      RSUB      RETURN TO CALLER
250      OUTPUT      BYTE      X'05'      CODE FOR OUTPUT DEVICE
255      .      END      FIRST
```

# Benefits of SIC/XE Addressing Modes

- Register-to-register instructions
  - Shorter than register-to-memory instructions
  - No memory reference
- Immediate addressing mode
  - No memory reference. The operand is already present as part of the instruction
- Indirect addressing mode
  - Avoids the needs for another instruction
- Relative addressing mode
  - Shorter than the extended instruction
  - Easy program relocation

# Considering Instruction Formats

- START directive specifies a beginning program address of 0: a relocatable program.
- Register-to-register instructions: simply convert the mnemonic name to their number equivalents
  - OPTAB: for opcodes
  - SYMTAB: preloaded with register names and their values

# Considering Instruction Formats

- COMPR A,S

-----

1010 0000 0000 0100 → A004

- CLEAR X

1011 0100 0001 0000 → B410

Mnemonic	Number	Special use
A	0	Accumulator; used for arithmetic operations
X	1	Ir
L	2	L
PC	8	P
SW	9	S

Mnemonic	Number	Special use
B	3	Base register; used for addressing
S	4	General working register—no special use
T	5	General working <u>register</u> —no special use
F	6	Floating-point accumulator (48 bits)

# Considering Addressing Modes

- PC or base relative addressing
  - Calculate displacement
  - Displacement must be small enough to fit in the 12-bit field (-2048..2047 for PC relative mode, 0..4095 for base relative mode)
- Extended instruction format (4-byte)
  - 20-bit field for direct addressing



# How Assembler Recognizes the Addressing Mode

- Extended format: +op m
- Indirect addressing: op @m
- Immediate addressing: op #c
- Index addressing: op m,X
- Relative addressing: op m
  - 1st choice: PC relative (arbitrarily chosen)
  - 2nd choice: base relative (if displacement is invalid in PC relative mode)
  - 3rd choice: error message (if displacement is invalid in both relative modes)

# SIC/XE Assembly with Object Code(fig.2.6)

Line	Loc	Source statement			Object code
5	0000	COPY	START	0	
10	0000	FIRST	<u>STL</u>	<u>RETADR</u>	17202D
12	0003		LDB	#LENGTH	69202D
13			BASE	LENGTH	
15	0006	CLOOP	<u>+JSUB</u>	<u>RDREC</u>	4B101036
20	000A		LDA	LENGTH	032026
25	000D		COMP	#0	290000
30	0010		JEQ	ENDFIL	332007
35	0013		<u>+JSUB</u>	<u>WRREC</u>	4B10105D
40	0017		<u>J</u>	<u>CLOOP</u>	3F2FEC
45	001A	ENDFIL	LDA	EOF	032010
50	001D		STA	BUFFER	0F2016
55	0020		LDA	#3	010003
60	0023		STA	LENGTH	0F200D
65	0026		<u>+JSUB</u>	<u>WRREC</u>	4B10105D
70	002A		J	@RETADR	3E2003
80	002D	EOF	BYTE	C'EOF'	454F46
95	0030	RETADR	RESW	1	
100	0033	LENGTH	RESW	1	
105	0036	BUFFER	RESB	4096	

# SIC/XE Assembly with Object Code

```
110      .
115      .          SUBROUTINE TO READ RECORD INTO BUFFER
120      .
125      1036      RDREC      CLEAR      X          B410
130      1038      CLEAR      A          B400
132      103A      CLEAR      S          B440
133      103C      +LDT      #4096      75101000
135      1040      RLOOP      TD          INPUT      E32019
140      1043      JEQ        RLOOP      332FFA
145      1046      RD          INPUT      DB2013
150      1049      COMPR      A,S        A004
155      104B      JEQ        EXIT      332008
160      104E      STCH      BUFFER,X      57C003
165      1051      TIXR      T          B850
170      1053      JLT        RLOOP      3B2FEA
175      1056      EXIT      STX          LENGTH      134000
180      1059      RSUB
185      105C      INPUT      BYTE      X'F1'      F1
195
```

# SIC/XE Assembly with Object Code

```
195      .
200      .          SUBROUTINE TO WRITE RECORD FROM BUFFER
205      .
210      105D      WRREC      CLEAR      X          B410
212      105F      LDT        LENGTH      774000
215      1062      WLOOP      TD          OUTPUT      E32011
220      1065      JEQ        WLOOP      332FFA
225      1068      LDCH       BUFFER,X      53C003
230      106B      WD          OUTPUT      DF2008
235      106E      TIXR       T          B850
240      1070      JLT        WLOOP      3B2FEF
245      1073      RSUB
250      1076      OUTPUT     BYTE      X'05'      05
255      END          FIRST
```

# SIC/XE Instruction Set

Mnemonic	Format	Opcode	Effect	Notes
ADD m	3/4	18	$A \leftarrow (A) + (m..m+2)$	
ADDF m	3/4	58	$F \leftarrow (F) + (m..m+5)$	X F
ADDR r1,r2	2	90	$r2 \leftarrow (r2) + (r1)$	X
AND m	3/4	40	$A \leftarrow (A) \& (m..m+2)$	
CLEAR r1	2	B4	$r1 \leftarrow 0$	X
COMP m	3/4	28	$(A) : (m..m+2)$	C
COMPF m	3/4	88	$(F) : (m..m+5)$	X F C
COMPR r1,r2	2	A0	$(r1) : (r2)$	X C
DIV m	3/4	24	$A \leftarrow (A) / (m..m+2)$	
DIVF m	3/4	64	$F \leftarrow (F) / (m..m+5)$	X F
DIVR r1,r2	2	9C	$r2 \leftarrow (r2) / (r1)$	X
FIX	1	C4	$A \leftarrow (F)$ [convert to integer]	X F
FLOAT	1	C0	$F \leftarrow (A)$ [convert to floating]	X F
HIO	1	F4	Halt I/O channel number (A)	P X

X: only for XE

C: set CC

F: floating-point

P: privileged

J m	3/4	3C	PC ← m	
JEQ m	3/4	30	PC ← m if CC set to =	
JGT m	3/4	34	PC ← m if CC set to >	
JLT m	3/4	38	PC ← m if CC set to <	
JSUB m	3/4	48	L ← (PC); PC ← m	
LDA m	3/4	00	A ← (m..m+2)	
LDB m	3/4	68	B ← (m..m+2)	X
LDCH m	3/4	50	A [rightmost byte] ← (m)	
LDF m	3/4	70	F ← (m..m+5)	X F
LDL m	3/4	08	L ← (m..m+2)	
LDS m	3/4	6C	S ← (m..m+2)	X
LDT m	3/4	74	T ← (m..m+2)	X
LDX m	3/4	04	X ← (m..m+2)	
LPS m	3/4	D0	Load processor status from information beginning at address m (see Section 6.2.1)	P X
MUL m	3/4	20	A ← (A) * (m..m+2)	

for interrupt



Mnemonic	Format	Opcode	Effect	Notes
MULF m	3/4	60	$F \leftarrow (F) * (m..m+5)$	X F
MULR r1, r2	2	98	$r2 \leftarrow (r2) * (r1)$	X
NORM	1	C8	$F \leftarrow (F)$ [normalized]	X F
OR m	3/4	44	$A \leftarrow (A) \mid (m..m+2)$	
RD m	3/4	D8	$A$ [rightmost byte] $\leftarrow$ data from device specified by (m)	P
RMO r1,r2	2	AC	$r2 \leftarrow (r1)$	X
RSUB	3/4	4C	$PC \leftarrow (L)$	
SHIFTL r1,n	2	A4	$r1 \leftarrow (r1)$ ; left circular shift n bits. {In assembled instruction, $r2 = n-1$ }	X
SHIFTR r1,n	2	A8	$r1 \leftarrow (r1)$ ; right shift n bits, with vacated bit positions set equal to leftmost bit of (r1). {In assembled instruction, $r2 = n-1$ }	X
SIO	1	F0	Start I/O channel number (A); address of channel program is given by (S)	P X


SSK m	3/4	EC	Protection key for address m	P X
Set Storage Key for memory protection			$\leftarrow (A)$ (see Section 6.2.4)	
STA m	3/4	0C	$m..m+2 \leftarrow (A)$	
STB m	3/4	78	$m..m+2 \leftarrow (B)$	X
STCH m	3/4	54	$m \leftarrow (A)$ [rightmost byte]	
STF m	3/4	80	$m..m+5 \leftarrow (F)$	X F
STI m	3/4	D4	Interval timer value $\leftarrow$ ( $m..m+2$ ) (see Section 6.2.1)	P X
STL m	3/4	14	$m..m+2 \leftarrow (L)$	
STS m	3/4	7C	$m..m+2 \leftarrow (S)$	X
STSW m	3/4	E8	$m..m+2 \leftarrow (SW)$	P
STT m	3/4	84	$m..m+2 \leftarrow (T)$	X
STX m	3/4	10	$m..m+2 \leftarrow (X)$	
SUB m	3/4	1C	$A \leftarrow (A) - (m..m+2)$	
SUBF m	3/4	5C	$F \leftarrow (F) - (m..m+5)$	X F



Mnemonic	Format	Opcode	Effect	Notes
SUBR r1,r2	2	94	$r2 \leftarrow (r2) - (r1)$	X
SVC n	2	B0	Generate SVC interrupt. {In assembled instruction, $r1 = n$ }	X
TD m	3/4	E0	Test device specified by (m)	P C
TIO	1	F8	Test I/O channel number (A)	P X C
TIX m	3/4	2C	$X \leftarrow (X) + 1$ ; (X): (m..m+2)	C
TIXR r1	2	B8	$X \leftarrow (X) + 1$ ; (X): (r1)	X C
WD m	3/4	DC	Device specified by (m) $\leftarrow$ (A) [rightmost byte]	P


# Immediate Addressing Mode

Instruction: 55 0020 LDA #3 010003



$(00)_{16}$  0 1 0 0 0 0  $(003)_{16}$   
 $(01)_{16}$   $(0)_{16}$   $(003)_{16}$

Instruction: 133 103C +LDT #4096 75101000



$(74)_{16}$  0 1 0 0 0 1  $(01000)_{16}$   
 $(75)_{16}$   $(1)_{16}$   $(01000)_{16}$

# Immediate Addressing Mode

- LDA #3

---- -- ni xbpe ----

0000 0001 0000 0000 0000 0011 → 010003

- +LDT #4096


---- -- ni xbpe ----

0111 0101 0001 0000 0001 0000 0000 0000

→ 75101000

# Extended Format

Instruction: 15 0006 CLOOP +JSUB RDREC 4B101036



$(48)_{16}$     1 1 0 0 0 1  

  
 $(4B)_{16}$      $(1)_{16}$

$(01036)_{16}$

$(01036)_{16}$


# Extended Format

- +JSUB RDREC

---- -- ni xbpe ---- ---- ---- ----  
0100 1011 0001 0000 0001 0000 0011 0110  
➔ 4B101036

# PC Relative Addressing Mode

Instruction: 10 0000 FIRST STL RETADR 17202D  
 12 0003 LDB #LENGTH 69202D  
 :  
 95 0030 RETADR RESW 1



PC is advanced **after** each instruction is fetched and **before** it is executed. That is, PC contains the address of the **next** instruction.

$$\text{disp} = (0030)_{16} - (0003)_{16} = (002D)_{16}$$

# PC Relative Addressing Mode

- STL RETADR

- - - - - - - ni xbpe ---- ---- ----


0001 0111 0010 ---- ---- ----

disp=RETADR-PC=0030-0003=002D

0001 0111 0010 0000 0010 1101 → 17202D

# PC Relative Addressing Mode

Instruction:    15    0006    CLOOP +JSUB    RDREC    4B101036  
                   :  
                   40    0017                    J    CLOOP    3F2FEC  
                   45    001A    ENDFIL    LDA    EOF    032010



$$\text{disp} = (0006)_{16} - (001A)_{16} = (-14)_{16} = (\text{FEC})_{16}$$



# PC Relative Addressing Mode

- J CLOOP

---- -- ni xbpe ---- ----

0011 1111 0010 ---- ----

disp=CLOOP-PC=006-01A=FEC= $(-14)_{16}$

0011 1111 0010 1111 1110 1100 → 3F2FEC

# PC Relative Addressing Mode

- J CLOOP

---- -- ni xbp0 ----

0011 1111 0bp0 ----


CLOOP-PC=0006-001A=FEC

$$\begin{array}{r}
 0000\ 0001\ 0100 = (14)_{16} = (20)_{10} \\
 +\ 1111\ 1110\ 1100 = (-14)_{16} = (-20)_{10} \\
 \hline
 0000\ 0000\ 0000 = (0)_{16} = (0)_{10}
 \end{array}$$

0011 1111 0010 1111 1110 1100 → 3F2FEC

Decimal	Signed-magnitude	Signed-1's complement	Signed-2's complement
+7	0111	0111	0111
+6	0110	0110	0110
+5	0101	0101	0101
+4	0100	0100	0100
+3	0011	0011	0011
+2	0010	0010	0010
+1	0001	0001	0001
+0	0000	0000	0000
-0	1000	1111	X
-1	1001	1110	1111
-2	1010	1101	1110
-3	1011	1100	1101
-4	1100	1011	1100
-5	1101	1010	1011
-6	1110	1001	1010
-7	1111	1000	1001
-8	X	X	1000

# Base Relative Addressing Mode



- PC relative is no longer applicable
- BASE directive **explicitly** informs the **assembler** that the base register will contain the address of LENGTH (use NOBASE to invalidate)
- LDB loads the address of LENGTH into base register during **execution**

$$\text{disp} = (0036)_{16} - (0033)_{16} = (0003)_{16}$$

# Base Relative Addressing Mode

- STCH BUFFER,X

---- -- ni xbp0 ---- ----

0101 0111 1bp0 ---- ----

BUFFER-PC=0036-1051


→ PC-relative fails

BUFFER-BASE=0036-0033=003

0101 0111 1100 0000 0000 0011 → 57C003

# Immediate + PC Relative Addressing Mode


Instruction: 12      0003                      LDB    #LENGTH              69202D  
                  13                                      BASE    LENGTH  
                  15      0006    CLOOP +JSUB    RDREC              4B101036  
                  :                                      :  
                  100      0033    LENGTH    RESW    1



$$\text{disp} = (0033)_{16} - (0006)_{16} = (002D)_{16}$$

# Indirect + PC Relative Addressing Mode

Instruction: 70      002A                      J      @RETADR                      3E2003  
                  80      **002D**      EOF                      BYTE      C'EOF'                      454F46  
                  95      0030      RETADR      RESW      1



$$\text{disp} = (0030)_{16} - (002D)_{16} = (0003)_{16}$$

# Why Program Relocation

- To increase the productivity of the machine



- Want to load and run several programs at the same time (multiprogramming)



- Must be able to load programs into memory wherever there is room



- Actual starting address of the program is not known until load time

# Absolute Program

- Program with starting address specified at assembly time
- In the example of SIC assembly program

Instruction:    55       101B                            LDA    THREE                            00102D


Calculated from the  
starting address 1000



- The address may be invalid if the program is loaded into some where else.



# Relocatable Program



# What Needs to be Relocated

- Need to be modified:
  - The **address portion** of those instructions that use **absolute** (direct) addresses.
- Need not be modified:
  - Register-to-register instructions (no memory references)
  - PC or base-relative addressing (relative displacement remains the same regardless of different starting addresses)

# How to Relocate Addresses

- For Assembler
  - For an address label, its address is assigned relative to the **start** of the program (that's why START 0)
  - Produce a modification record to store the starting location and the length of the address field to be modified.
- For loader
  - For each modification record, **add** the **actual** beginning address of the program to the address field at load time.

# Format of Modification Record

Modification record:

Col. 1	M
Col. 2–7	Starting location of the address field to be modified, relative to the beginning of the program (hexadecimal)
Col. 8–9	Length of the address field to be modified, in half-bytes (hexadecimal)

- One modification record for each address to be modified
- The length is stored in half-bytes (**20 bits = 5 half-bytes**)
- The starting location is the location of the byte containing **the leftmost bits** of the address field to be modified.
- If the field contains **an odd number of half-bytes**, the starting location begins in **the middle of the first byte**.

# Relocatable Object Program

HCOPY 000000001077

0 1 2 3 4 5 6 7

5 half-bytes

T0000001D17202D69202D4B1010360320262900003320074B10105D3F2FEC032010

T00001D130F20160100030F200D4B10105D3E2003454F46

T0010361DB410B400B44075101000E32019332FFADB2013A00433200857C003B850

T0010531D3B2FEA1340004F0000F1B410774000E32011332FFA53C003DF2008B850

T001070073B2FEF4F000005

M00000705

M00001405

M00002705

E000000

15 +JSUB RDREC

35 +JSUB WRREC

65 +JSUB WRREC